<commit_before>{ stdenv, fetchurl, pkgconfig, libusb1
, qtbase, qttools, makeWrapper, qmake
, withEspeak ? false, espeak ? null }:

stdenv.mkDerivation  rec {
  name = "rockbox-utility-${version}";
  version = "1.4.0";

  src = fetchurl {
    url = "http://download.rockbox.org/rbutil/source/RockboxUtility-v${version}-src.tar.bz2";
    sha256 = "0k3ycga3b0jnj13whwiip2l0gx32l50pnbh7kfima87nq65aaa5w";
  };

  buildInputs = [ libusb1 qtbase qttools ]
    ++ stdenv.lib.optional withEspeak espeak;
  nativeBuildInputs = [ makeWrapper pkgconfig qmake ];

  preConfigure = ''
    cd rbutil/rbutilqt
  '';

  installPhase = ''
    runHook preInstall

    install -Dm755 RockboxUtility $out/bin/rockboxutility
    ln -s $out/bin/rockboxutility $out/bin/RockboxUtility
    wrapProgram $out/bin/rockboxutility \
    ${stdenv.lib.optionalString withEspeak ''
      --prefix PATH : ${espeak}/bin
    ''}

    runHook postInstall
  '';

  meta = with stdenv.lib; {
    description = "Open source firmware for mp3 players";
    homepage = http://www.rockbox.org;
    license = licenses.gpl2;
    platforms = platforms.linux;
    maintainers = with maintainers; [ goibhniu jgeerds ];
  };
}<commit_msg>rockbox_utility: fix path to lrelease<commit_after>{ stdenv, fetchurl, pkgconfig, libusb1
, qtbase, qttools, makeWrapper, qmake
, withEspeak ? false, espeak ? null }:

let inherit (stdenv.lib) getDev; in

stdenv.mkDerivation  rec {
  name = "rockbox-utility-${version}";
  version = "1.4.0";

  src = fetchurl {
    url = "http://download.rockbox.org/rbutil/source/RockboxUtility-v${version}-src.tar.bz2";
    sha256 = "0k3ycga3b0jnj13whwiip2l0gx32l50pnbh7kfima87nq65aaa5w";
  };

  buildInputs = [ libusb1 qtbase qttools ]
    ++ stdenv.lib.optional withEspeak espeak;
  nativeBuildInputs = [ makeWrapper pkgconfig qmake ];

  postPatch = ''
    sed -i rbutil/rbutilqt/rbutilqt.pro \
        -e '/^lrelease.commands =/ s|$$\[QT_INSTALL_BINS\]/lrelease -silent|${getDev qttools}/bin/lrelease|'
  '';

  preConfigure = ''
    cd rbutil/rbutilqt
  '';

  installPhase = ''
    runHook preInstall

    install -Dm755 RockboxUtility $out/bin/rockboxutility
    ln -s $out/bin/rockboxutility $out/bin/RockboxUtility
    wrapProgram $out/bin/rockboxutility \
    ${stdenv.lib.optionalString withEspeak ''
      --prefix PATH : ${espeak}/bin
    ''}

    runHook postInstall
  '';

  meta = with stdenv.lib; {
    description = "Open source firmware for mp3 players";
    homepage = http://www.rockbox.org;
    license = licenses.gpl2;
    platforms = platforms.linux;
    maintainers = with maintainers; [ goibhniu jgeerds ];
  };
}
{'fraction_non_alphanumeric': 0.12691131498470948, 'fraction_numerical': 0.024464831804281346, 'mean_word_length': 3.4506802721088436, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 2, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 4, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '7048560', 'n_tokens_mistral': 1022, 'n_tokens_neox': 982, 'n_words': 178}